

Binary IFA-IRT Models in Mplus version 8.1

Example data: 635 older adults (age 80-100) self-reporting on 7 items assessing the Instrumental Activities of Daily Living (IADL) as follows:

1. Housework (cleaning and laundry): 1=64%
2. Bedmaking: 1=84%
3. Cooking: 1=77%
4. Everyday shopping: 1=66%
5. Getting to places outside of walking distance: 1=65%
6. Handling banking and other business: 1=73%
7. Using the telephone 1=94%

Two versions of a response format were available:

Binary → 0 = “needs help”, 1 = “does not need help”

Categorical → 0 = “can’t do it”, 1=“big problems”, 2=“some problems”, 3=“no problems”

Higher scores indicate greater function. We will look at each response format in turn.

Binary 2-PL Model Syntax (left) and 1-PL Model Syntax (right) using ML and a logit scale:

```

TITLE: Assess binary IADL items using 2PL
DATA: FILE IS ADL.dat;

VARIABLE: NAMES ARE case dial-dia7 cial-cia7;
              USEVARIABLES ARE dial-dia7;
              CATEGORICAL ARE dial-dia7;
              MISSING ARE .;
              IDVARIABLE IS case;

ANALYSIS: ESTIMATOR IS ML;
              LINK IS LOGIT;

MODEL:
! Factor loadings all estimated in 2PL
  IADL BY dial-dia7*;
! Item thresholds all estimated
  [dial$1-dia7$1*];
! Factor mean=0 and variance=1 for identification
  [IADL@0]; IADL@1;

OUTPUT:      STDYX;                ! Standardized solution
              RESIDUAL TECH10;       ! Local fit info

SAVEDATA:   SAVE = FSCORES;        ! Save factor scores (thetas)
              FILE = IADL_2PLThetas.dat; ! File factor scores saved to

PLOT:      TYPE IS PLOT1;         ! PLOT1 gets you sample descriptives
              TYPE IS PLOT2;         ! PLOT2 gets you the IRT-relevant curves
              TYPE IS PLOT3;         ! PLOT3 gets you descriptives for theta

```

```

TITLE: Assess binary IADL items using 1PL
DATA: FILE IS ADL.dat;

VARIABLE: NAMES ARE case dial-dia7 cial-cia7;
              USEVARIABLES ARE dial-dia7;
              CATEGORICAL ARE dial-dia7;
              MISSING ARE .;
              IDVARIABLE IS case;

ANALYSIS: ESTIMATOR IS ML;
              LINK IS LOGIT;

MODEL:
! Factor loadings all held equal in 1PL
  IADL BY dial-dia7* (loading);
! Item thresholds all estimated
  [dial$1-dia7$1*];
! Factor mean=0 and variance=1 for identification
  [IADL@0]; IADL@1;

OUTPUT:      STDYX;                ! Standardized solution
              RESIDUAL TECH10;       ! Local fit info

SAVEDATA:   SAVE = FSCORES;        ! Save factor scores (thetas)
              FILE = IADL_1PLThetas.dat; ! File factor scores saved to

PLOT:      TYPE IS PLOT1;         ! PLOT1 gets you sample descriptives
              TYPE IS PLOT2;         ! PLOT2 gets you the IRT-relevant curves
              TYPE IS PLOT3;         ! PLOT3 gets you descriptives for theta

```

Binary 2-PL Model Fit (left) and 1-PL Model Fit (right) using ML logit:



MODEL FIT INFORMATION - 2PL	MODEL FIT INFORMATION - 1 PL		
Number of Free Parameters	14	Number of Free Parameters	8
Loglikelihood		Loglikelihood	
H0 Value	-1454.634	H0 Value	-1464.457
Information Criteria		Information Criteria	
Akaike (AIC)	2937.268	Akaike (AIC)	2944.915
Bayesian (BIC)	2999.619	Bayesian (BIC)	2980.544
Sample-Size Adjusted BIC	2955.170	Sample-Size Adjusted BIC	2955.144
(n* = (n + 2) / 24)		(n* = (n + 2) / 24)	
Chi-Square Test of Model Fit for the Binary and Ordered Categorical (Ordinal) Outcomes		Chi-Square Test of Model Fit for the Binary and Ordered Categorical (Ordinal) Outcomes**	
Pearson Chi-Square		Pearson Chi-Square	
Value	340.829	Value	296.199
Degrees of Freedom	113	Degrees of Freedom	118
P-Value	0.0000	P-Value	0.0000
Likelihood Ratio Chi-Square		Likelihood Ratio Chi-Square	
Value	120.273	Value	126.354
Degrees of Freedom	113	Degrees of Freedom	118
P-Value	0.3023	P-Value	0.2828
Linda Muthén suggests that if these 2 χ^2 values don't match, they should not be used to assess model fit.		** Of the 630 cells in the latent class indicator table, 1 were deleted in the calculation of chi-square due to extreme values.	
Further, the possible total df for the χ^2 is calculated based on # possible response patterns. Here, for 7 binary items:		This error message indicates that these 2 sets of chi-squares for the 2-PL and 1-PL are not on the same scale because they are not based on the same data. So we can't compare the chi-squares to test the difference in model fit, but we can still compare LL values.	
2PL model: $2^7 = 128$ possible – 7 loadings – 7 thresholds – 1 = 113			
1PL model: $2^7 = 128$ possible – 1 loading – 7 thresholds – 1 = 119			
However, the 1PL only has df=118 because of the deleted cell.			

Does the 2-PL fit better than the 1-PL?



-1454.634*-2 = 2909.258 -2LL difference = 19.946, df = 6, p = .0032
-1464.457*-2 = 2928.914 AIC (but not BIC) is smaller for 2PL, too

3 differently scaled 2-PL solutions from ML logit provided by Mplus – all provide the exact same model predictions!


UNSTANDARDIZED MODEL RESULTS (IFA MODEL SOLUTION)					IRT PARAMETERIZATION IN TWO-PARAMETER LOGISTIC METRIC WHERE THE LOGIT IS DISCRIMINATION*(THETA - DIFFICULTY)				
		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value				
FACTOR LOADINGS = CHANGE IN LOGIT(Y=1) PER UNIT CHANGE IN THETA					Item Discriminations = SLOPE OF ICC AT P=.50				
IADL	BY					IADL	BY		
DIA1		4.328	0.560	7.725	0.000	DIA1		4.328	0.560
DIA2		4.978	0.808	6.159	0.000	DIA2		4.978	0.808
DIA3		4.323	0.570	7.579	0.000	DIA3		4.323	0.570
DIA4		7.511	1.696	4.429	0.000	DIA4		7.511	1.696
DIA5		4.248	0.527	8.062	0.000	DIA5		4.248	0.527
DIA6		3.451	0.401	8.600	0.000	DIA6		3.451	0.401
DIA7		3.283	0.601	5.467	0.000	DIA7		3.283	0.601
THRESHOLDS = EXPECTED LOGIT(Y=0) WHEN THETA IS 0					Item Difficulties = LOCATION OF ITEM ON LATENT TRAIT at P=.50, LOGIT=0				
DIA1\$1		-1.629	0.295	-5.516	0.000	DIA1\$1		-0.376	0.052
DIA2\$1		-5.202	0.770	-6.754	0.000	DIA2\$1		-1.045	0.065
DIA3\$1		-3.462	0.441	-7.842	0.000	DIA3\$1		-0.801	0.059
DIA4\$1		-3.120	0.744	-4.193	0.000	DIA4\$1		-0.415	0.047
DIA5\$1		-1.833	0.298	-6.158	0.000	DIA5\$1		-0.432	0.052
DIA6\$1		-2.442	0.292	-8.368	0.000	DIA6\$1		-0.708	0.060
DIA7\$1		-5.962	0.858	-6.951	0.000	DIA7\$1		-1.816	0.126
STDYX MODEL RESULTS (STANDARDIZED IFA MODEL SOLUTION)					USING RESULTS FROM IFA MODEL (LEFT PANEL):				
		Estimate	S.E.	Est./S.E.	Two-Tailed P-Value	IFA model: $\text{Logit}(y) = -\text{threshold} + \text{loading}(\text{Theta})$			
FACTOR LOADINGS IN STANDARDIZED METRIC = $\text{loading} \cdot \text{SD}(\text{Theta}) / \text{SD}(Y)$					Threshold = expected logit of (y=0) for someone with Theta=0				
IADL	BY					When *-1, threshold becomes intercept: expected logit for (y=1) instead			
DIA1		0.922	0.018	51.712	0.000	Loading = regression of item logit on Theta			
DIA2		0.940	0.018	52.557	0.000	= change in logit(y) for a one-unit change in Theta			
DIA3		0.922	0.018	50.622	0.000	IFA Models:			
DIA4		0.972	0.012	80.380	0.000	Logit (DIA1=1) = 1.629 + 4.328(Theta) → if Theta=0, prob(y=1)= .836			
DIA5		0.920	0.018	52.291	0.000	Logit (DIA7=1) = 5.962 + 3.283(Theta) → if Theta=0, prob(y=1)= .997			
DIA6		0.885	0.022	39.729	0.000				
DIA7		0.875	0.037	23.380	0.000				
THRESHOLDS IN STANDARDIZED METRIC = $\text{threshold} / \text{SD}(Y)$					USING RESULTS FROM IRT MODEL (RIGHT PANEL):				
DIA1\$1		-0.347	0.048	-7.303	0.000	IRT model: $\text{Logit}(y=1) = a(\text{theta} - \text{difficulty})$			
DIA2\$1		-0.982	0.056	-17.409	0.000	a = discrimination (rescaled slope) = loading/1.7			
DIA3\$1		-0.739	0.051	-14.373	0.000	b = difficulty (location on latent metric) = threshold/loading			
DIA4\$1		-0.404	0.045	-8.928	0.000				
DIA5\$1		-0.397	0.048	-8.348	0.000				
DIA6\$1		-0.626	0.050	-12.558	0.000				
DIA7\$1		-1.590	0.080	-19.949	0.000				
R-SQUARE = standardized loading²					IRT Models:				
DIA1		0.851	0.033	25.856	0.000	Logit (DIA1=1) = 4.328*(Theta - -0.376) → if Theta=0, prob(y=1)= .836			
DIA2		0.883	0.034	26.278	0.000	Logit (DIA7=1) = 3.283*(Theta - -1.816) → if Theta=0, prob(y=1)= .997			
DIA3		0.850	0.034	25.311	0.000				
DIA4		0.945	0.024	40.190	0.000				
DIA5		0.846	0.032	26.145	0.000				
DIA6		0.784	0.039	19.865	0.000				
DIA7		0.766	0.066	11.690	0.000				



Distribution of Theta under 2 PL (made in Mplus): Although reliability is > .80 from -1.5 to 0.3 or so, we see a huge ceiling effect: most respondents can do all the tasks.



The estimated theta scores are supposed to have a mean of 0 and a variance of 1, but this table shows that they have a variance of only .741 instead. Such shrinkage is why it can be problematic to use these estimated theta scores as observed variables in other analyses.




Plots of item parameters and predicted probabilities of item responses (made in excel):

Top Left: Note that no items are available to measure above-average abilities well! The item difficulty for most items covers values of Theta between -1.0 to -0.5 .


Bottom Left: These are the thresholds for each item, or the logit of $(y=0)$ if $\text{Theta}=0$. These are hard to interpret as is....

Bottom Right: These are the probability of $y=1$ if $\text{Theta}=0$, as given by $1 - [\exp(\text{threshold}) / (1 + \exp(\text{threshold}))]$

See excel workbook for calculations and plots



These are the implied probabilities of $(y = 1)$ for a person with $\text{Theta} = 0$



Here is another estimation approach: a 2PL vs. a 1PL for Binary Responses using WLSMV Probit model

<pre> TITLE: 2PL Binary Model under WLSMV DATA: FILE IS ADL.dat; VARIABLE: NAMES ARE case dial-dia7 cial-cia7; USEVARIABLES ARE dial-dia7; CATEGORICAL ARE dial-dia7; MISSING ARE .; IDVARIABLE IS case; ANALYSIS: ESTIMATOR IS WLSMV; PARAMETERIZATION IS THETA; MODEL: ! Factor loadings all estimated in 2PL IADL BY dial-dia7*; ! Item thresholds all estimated [dial\$1-dia7\$1*]; ! Factor mean=0 and variance=1 for identification [IADL@0]; IADL@1; OUTPUT: STDYX Residual; ! Standardized solution, local fit SAVEDATA: DIFFTEST=2PL.dat; ! Save info from bigger model SAVE = FSCORES; ! Save factor scores (thetas) FILE IS IADL_2PLThetas.dat; ! File factor scores saved to PLOT: TYPE IS PLOT1 PLOT2 PLOT3; ! Get IRT plots MODEL FIT INFORMATION Number of Free Parameters 14 Chi-Square Test of Model Fit Value 54.820* Degrees of Freedom 14 P-Value 0.0000 * The chi-square value for MLM, MLMV, MLR, ULSMV, WLSM and WLSMV cannot be used for chi-square difference testing in the regular way. MLM, MLR and WLSM chi-square difference testing is described on the Mplus website. MLMV, WLSMV, and ULSMV difference testing is done using the DIFFTEST option. RMSEA (Root Mean Square Error Of Approximation) Estimate 0.068 90 Percent C.I. 0.049 0.087 Probability RMSEA <= .05 0.055 CFI/TLI CFI 0.997 TLI 0.995 Chi-Square Test of Model Fit for the Baseline Model Value 12351.798 Degrees of Freedom 21 P-Value 0.0000 WRMR (Weighted Root Mean Square Residual) Value 1.160 </pre>	<pre> TITLE: 1PL Binary Model under WLSMV DATA: FILE IS ADL.dat; VARIABLE: NAMES ARE case dial-dia7 cial-cia7; USEVARIABLES ARE dial-dia7; CATEGORICAL ARE dial-dia7; MISSING ARE .; IDVARIABLE IS case; ANALYSIS: ESTIMATOR IS WLSMV; PARAMETERIZATION IS THETA; DIFFTEST=2PL.dat; ! Use saved info from bigger model MODEL: ! Factor loadings all equal in 1PL IADL BY dial-dia7* (loading); ! Item thresholds all estimated [dial\$1-dia7\$1*]; ! Factor mean=0 and variance=1 for identification [IADL@0]; IADL@1; OUTPUT: STDYX Residual; ! Standardized solution, local fit SAVEDATA: SAVE = FSCORES; ! Save factor scores (thetas) FILE IS IADL_1PLThetas.dat; ! File factor scores saved to PLOT: TYPE IS PLOT1 PLOT2 PLOT3; ! Get IRT plots MODEL FIT INFORMATION Number of Free Parameters 8 Chi-Square Test of Model Fit Value 64.889* Degrees of Freedom 20 P-Value 0.0000 Chi-Square Test for Difference Testing Value 17.874 Degrees of Freedom 6 P-Value 0.0066 RMSEA (Root Mean Square Error Of Approximation) Estimate 0.059 90 Percent C.I. 0.044 0.076 Probability RMSEA <= .05 0.154 CFI/TLI CFI 0.996 TLI 0.996 SRMR (Standardized Root Mean Square Residual) Value 0.056 The Chi-Square for Difference Testing tells us directly that the 2PL version of the binary model fits significantly better (now under WLSMV, same as it did under ML). </pre>
---	--

Extensive Results Section (in which model fit via WLSMV is reported first, followed by full-information MML as “better” version of model parameters). Note this is *way* more text than one would typically write, but I provide it here for completeness:

Psychometric assessment for the extent to which a single latent trait could predict that pattern of association among these 7 binary items was conducted using Item Factor Analysis (IFA) in *Mplus* v 8.1 (Muthén and Muthén, 1998–2017). These models use a link function (i.e., logit or probit) and a conditional Bernoulli response distribution to predict the conditional probability of a response = 1 (instead of 0) from a linear model as $Link(y_{is} = 1) = -\tau_i + \lambda_i F_s$. In this item model, $-\tau_i$ is the negative of an item-specific threshold (which becomes an intercept when multiplied by -1) that gives the link-transformed probability of response $y_{is} = 1$ (for item i and subject s) at a latent trait score F for subject s of 0, and λ is a factor loading for the expected change in the link-transformed response for a one-unit change in F_s . No separate item-specific residual variances can be estimated given these items’ binary response options.

The current gold standard of estimation for IFA models is marginal maximum likelihood (MML), in which the term *marginal* refers to the full-information process of allowing all possible trait values for each person in the analysis using adaptive Gaussian quadrature with 15 points per factor. Accordingly, measures of model fit when using MML involve the contingency table of all possible responses to all items. In our 7 items, the full contingency table generates up to $2^7 = 128$ possible cells. Consequently, no measures of absolute fit would be valid for the current sample of 635 respondents (which would need a minimum expected count of 5 respondents within each possible cell). Instead, we conducted assessment of model fit via a limited-information diagonally weighted least squares estimator using a mean- and variance-corrected χ^2 (i.e., WLSMV in *Mplus* with the THETA parameterization and a probit link function). In the WLSMV estimator, the item responses are first summarized into an estimated tetrachoric correlation matrix using the cross-tabulation of responses for each possible pair of items. The IFA models are then fitted to the estimated polychoric correlation matrix, such that traditional measures of global and local absolute fit (i.e., traditional in confirmatory factor analyses of continuous responses) can be computed by comparing the model-predicted and data-estimated polychoric correlation matrices. In addition to χ^2 tests of absolute fit, it also provides the Comparative Fit Index (CFI), the Standardized Root Mean Square Residual (SRMR), and the Root Mean Square Error of Approximation (RMSEA). The CFI indexes the fit of the specified model relative to a null model (of no tetrachoric correlations across items), in which CFI values $\geq .95$ indicate excellent fit. Conversely, the SRMR and RMSEA index the fit of the specified model relative to a saturated model (i.e., the data-estimated tetrachoric correlations), in which SRMR and RMSEA values $\leq .05$ indicate excellent fit. RMSEA also offers a 90% confidence interval and a significance test of “close fit” with a null hypothesis of $.05$. Local misfit can be diagnosed by examining the specific sources of discrepancy between the model-predicted and data-estimated tetrachoric correlations (i.e., as available using the RESIDUAL option in *Mplus*). Finally, the fit of nested models can be compared using the DIFFTEST procedure in *Mplus*.

A single-trait model was first fit for the seven binary items using WLSMV, in which the latent trait mean and variance were fixed for identification to 0 and 1, respectively, and separate thresholds and factor loadings were estimated for each item. This model exhibited acceptable fit by every measure except the χ^2 test of absolute fit, $\chi^2(14) = 54.820$, $p < .001$, CFI = .997, SRMR = .037, RMSEA = .068 [CI = .049–.087, $p = .055$]. Examination of local misfit revealed all discrepancies between the model-predicted and data-estimated tetrachoric correlations were less than .112 in absolute value, indicating no practically significant bivariate item misfit. A reduced model in which all loadings were constrained equal across items fit significantly worse, DIFFTEST(6) = 17.874, $p = .007$, indicating differences in item discrimination (i.e., the extent to which each item was related to the latent trait). Thus, the original model was retained for further examination using full-information marginal maximum likelihood (MML) estimation instead.

Model parameters obtained using MML and a logit link are shown in Table 1, which includes the IFA item parameters (thresholds and loadings), as well as their Item Response Theory (IRT) analogous parameter of item difficulty, computed as $b_i = \tau_i / \lambda_i$; IRT discrimination a_i is the same as the loading λ_i in this case. The net result of these item parameters can be described more succinctly by examining the overall reliability with which the latent trait has been measured. In IFA or IRT models—as in any kind of psychometric model with a nonlinear relationship between the item response and the latent trait—reliability is trait-specific, most often characterized by a quantity known as *test information*. For ease of interpretation, the test information function created by the items was converted to a traditional measure of reliability that ranges from 0 to 1 as $reliability = information / (information + 1)$. Figure 1 shows that test reliability is $\geq .80$ only from ~ 1.8 SD below the mean to 0.20 SD above the mean, after which point reliability drops off precipitously due to a lack of items with difficulty levels above 0.

(Table 1 would have all estimated IFA item parameters and their SEs, as well IRT parameters (and SEs if available); Figure 1 is in Example 5 spreadsheet)
References: Muthén, L. K., & Muthén, B.O. (1998–2017). *Mplus User’s Guide* (Eighth Edition). Los Angeles, CA: Muthén & Muthén.